
Figure 1.5. Figure 1.4

Notice that the box includes the items we had on our list. Dialogs can have up to 12 buttons at a time,
starting at the lower left and working their way to the right and upwards. When you click one of the buttons,
the dialog sends the text for that button to the channel you specified when you created it. In this case, that's
channel 587178742. Fortunately, that was easy to change from 42 because we used a single variable, so
now all of the command listeners are still listening in the right place. Just make sure you set the listener
to listen for any text, and not just the string "on".

Save the script and notice that the dialog appears. Try clicking the buttons. So far we've only implemented
the actions we already had, so all you can do is turn logging on and off, but we added the ability to bring
up the dialog box when the user clicks the object, in case they dismiss the dialog to get it out of the way.
We've used the llDetected(0) function to determine the key of the person who touched the logger so we
can display the dialog box for them.

It's important to understand that using dialogs is just like speaking outright on a channel; there's nothing
tying the dialog to the object. If you create a copy of the logger by editing it and holding shift while you
drag one if its position arrows, you'll see that a single dialog affects both objects, as in Figure 1.5:

Figure 1.6. Figure 1.5

Go ahead and delete the extra logger object to avoid confusion.

5.8. 1.4.8 Logging the chat
At this point it seems like we've done everything but actually log the chat, so let's go ahead and put that
in place.

Logging the chat is a pretty simple and straightforward process; we just need to create a global variable
and update it when necessary, as in Listing 1.8:

string log = "";
integer loggerhandle;
...
     listen(integer channel, string name, key id, string message)
    {
        if (channel == 0){
           
           log = log + name + ": " + message + "\n";
           
        } else if (channel == commandChannel) {
            if (message == "Logging Off"){
                llListenRemove(loggerhandle);
                state default;
            } else if (message == "Send Log"){
                 llOwnerSay(log);
                 log = "";
            }
        }
    }
...

The only time we want to update the log variable is when we're in the logging state and someone says
something to log, so we'll add that to the listen() event. Note that the message variable includes only

what was said, and not who said it. To compensate, we'll use the name variable to add that information
manually. Each time the script hears chat text it needs to log, it takes the current log string, adds the new
information to it, and then stores that new string in the log variable.

Tip

Because of the way LSL performs string (and list) concatenations, the way we're updating the log
here causes two copies of the log to exist in memory at one time. It's not a big deal now, but as you
get into situations where every byte of memory counts, you'll want to use a structure more like

log = (log = "") + log + name + ": " + message + "\n";

It's confusing, but basically it forces LSL to clear out the original value after it's made the internal
copy it uses for the actual concatenation.

If you save the script and try it out, the log will look something like this:

Chase Marellan: Can you hear me?
Tria Boucher: Yes, I can hear you just fine, what are you shouting for?
Chase Marellan:  Oh!  I didn't know someone else was here.  
Chase Marellan:  And I wasn't shouting, just check your history. :)

I've also added a little bit of code so that the object outputs the log string when you click "Send Log" so
that we can see what has happened, but that's far from a permanent solution! In the next section, we'll look
at periodically emailing the log so it doesn't get lost.

6. 1.5 Sending Email
Let's take stock of where we are. We've created an object and added a script that turns logging on and off.
When logging is on, it collects all of the text said around it, but at this point the logger is not doing anything
with that logged text. As soon as the script resets, the object is deleted or taken back into inventory, and
the record is gone. What we need is some way to persist the information, or make it more permanent -- or
at least available later! We'll do that by having the script periodically email the information home.

There was a time when email was the only way to get an object's information into or out of Second
Life. Fortunately, we now have HTTP -- which we'll cover in the future chapters -- but email remains
an important tool in your LSL arsenal. Objects can not only send email, but also receive it, making it a
convenient way to contact an object from the outside.

But let's start simply, with emailing the log to you so you'll have it later.

6.1. 1.5.1 Sending an email message
Sending an email message from LSL is a pretty straightforward proces, as you can see in Listing 1.9:

...
string log = "";

string logDestination = "your@email.here";

sendLog(string emailToSend, string textToSend){
    
   llEmail(emailToSend, "Chat Log", textToSend);

   log = "";
   llDialog(llGetOwner(), "Logging control", controlButtons, commandChannel);

}

default
{
...
     listen(integer channel, string name, key id, string message)
    {
       if (message == "Logging On"){
           state logging;
        } else if (message == "Send Log"){
                sendLog(logDestination, log);
                state logging;
            }
    }
}

state logging {
    
...
            } else if (message == "Send Log"){
                sendLog(logDestination, log);
            }
...

To make things simple, we've added a single function, sendLog(), that we can call from anywhere in the
script. This function takes in the recipient email address and the contentt to email, and sends it off with a
subject of "Chat Log". We can then call that function whenever the user clicks the "Send Log" button. If
you save the script and test it out, you'll notice two things. First, the script pauses for 20 seconds; this is
normal, and we'll deal with it in a moment. Second, you'll see an email something like this:

From: "Object" <87cdf7ed-e5ce-e84e-9b4b-4146da66ba37@lsl.secondlife.com>
To: <nchase@earthlink.net>
Subject: Chat Log

Object-Name: Object
Region: InterSection (212736, 272640)
Local-Position: (31, 223, 35)

Chase Marellan: This life is a test.
Chase Marellan: It is only a test.
Chase Marellan: Had this been a real life, do you think anybody would tell you?

Notice that it looks like a typical email, with the exception of identifying information added to the email
before it goes out. (So you can forget about those anonymous spammer plans!) The "From" line identifies
the object as, well, "Object", but you can change that by changing the object properties as you saw in
Chapter 2. For example, we can rename this object "Chat Logger".

The email address is a unique key that identifies the object, and as you'll see in the next section, can be
used to send email to the object.

Before we look at receiving email, however, we need to deal with some limitation built into object-based
emails.

6.2. 1.5.2 Email limitations
As you might imagine, a system that makes it easy to send emails is ripe for abuse, so there are certain
restrictions in place you need to keep in mind when using this system to send emails:

• Script delay: Once you fire off an email, the script pauses for 20 seconds. That means it will do nothing
during that time, including log more chat. This delay is per script, so we can get around that by using
multiple scripts, which we'll look at next.

• Email volume: If you send too many emails -- the exact definition of "too many" is up for debate and is
frequently revised -- your emails will simply disappear into a black hole, never to be seen again. They're
tracked by recipient, not script, and limits of 1 email per second or so have been quoted on the LSL Wiki.

• Email size: The entire email, including headers, can be no larger than 4096 characters. This means we're
going to have to check the size of the log and send and reset it before it gets too big, as in Listing 1.10:

...
     listen(integer channel, string name, key id, string message)
    {
        if (channel == 0){
           
           log = log + name + ": " + message + "\n";
           if (llStringLength(log) > 3000){
               sendLog(logDestination, log);
            }
           
        } else if (channel == commandChannel) {
...

Here it's a simple matter of checking the length of the log string using the llStringLength()
function.

As I said, the request can hold as much as 4096 bytes, but because that includes any headers sent with the
request, we're going to cap the body at approximately 3000 characters and send it off. The sendLog()
function takes care of clearning the log.

6.3. 1.5.3 Linking scripts
If you've been paying attention, you're probably worried about that 20 second delay that happens when
you send off an email. After all, it's one thing if you get a delay when you purposely send off the log, but to
have it suddenly stop in the middle of a conversation could be a real showstopper. Fortunately, that delay
is per script, and not per object. That means we can create a second script and send it the log to process.
The logger continues on its merry way, and we don't care if the sending script is temporarily delayed.

Start by right-clicking the original script in the Contents pane and choosing Rename to rename the script
"Main Logger Script". Next, click the New Script button to create a new script. Double-click the script to
open it and add the code shown in Listing 1.11:

string logDestination = "your@email.here";

default
{
    link_message(integer sender_num, integer num, string str, key id)
    {
        llOwnerSay("Sending log ...");
        llEmail(logDestination, "Chat Log", str);
    }
}

This script is just a normal script with one event handler, link_message. To get this event to fire, we
need to update the sendLog() function to call llMessageLinked() in the main script, as in Listing
1.12:

...
sendLog(string emailToSend, string textToSend){
   llMessageLinked(LINK_THIS, 0, textToSend, NULL_KEY);
   log = "";
   llDialog(llGetOwner(), "Logging control", controlButtons, commandChannel);
}
...

This function is normally used to send messages between prims that are linked together, but there's no
reason we can't use it to send a message within a prim, hence the use of the LINK_THIS constant to
identify the target. This function sends three values: an integer, a string, and a key value -- we'll talk about
those in the next chapter -- to any relevant object looking for the link_message() event.

In our case, it's the mail sender script.

So let's put it all together. Save the script, turn on logging and enter some chat text. Click the "Send Log"
button and notice that rather than waiting for 20 seconds, as it did before, it returns almost immediately,
and you can turn logging on again and type more text. The mail sender script tidily sends off the email
in the background.

This technique is crucial in the LSL environment, because you can use it to simulate "threads", where
the script goes about its business while time-consuming (or deliberately script-delayed) actions happen
in the background.

So that's it. No more lost meetings!

7. 1.6 Summary
In this chapter you learned the basic concepts that will enable you do make Linden Script Language, or
LSL, do just about whatever you need it to. You learned that scripts consists of event handlers, which
perform specific actions when specific events take place. You also learned that these event handlers can
be grouped into states, with one state active at any given time.

The ability to react to events enabled you to build a chat logger that reacts to the text it hears, and the use
of states enabled you to turn that logging action on and off. Putting all fo that together, you built a simple
logging device that listens on both public and private channels, and added a dialog box that makes it easier
for users to use. In addition, you learned to send emails using a script. Finally, you learned about some
of the limitations on LSL scripts, such as script delays designed to prevent abuse, and how to simulate
threading by using link messages.

In the next chapter, you'll learn more about using LSL, including receiving emails, using timers and giving
inventory. You'll do this as you create an update server that gives users the latest version of a product.

